﻿/**
* @private
* @license	GNU GPLv2
*
* Copyright (C) 2007  Polly Jex.
* 
* This program is free software; you can redistribute it and/or
* modify it under the terms of the GNU General Public License
* as published by the Free Software Foundation; either version 2
* of the License, or (at your option) any later version.
* 
* This program is distributed in the hope that it will be useful,
* but WITHOUT ANY WARRANTY; without even the implied warranty of
* MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
* GNU General Public License for more details.
* 
* You should have received a copy of the GNU General Public License
* along with this program; if not, write to the Free Software
* Foundation, Inc., 51 Franklin Street, Fifth Floor, Boston, 
* MA  02110-1301, USA.
*
* @see	http://www.gnu.org/licenses/old-licenses/gpl-2.0.html
*/

package com.kirika.data.tda {	

	/**
	 * @import
	 */
	
	import com.kirika.core.CoreObject;
	import com.kirika.data.IIterator;
	import com.kirika.utils.ClassUtil;
	import com.kirika.utils.NumberUtil;
	
	/**
	 * Modelo abstracto de Iteracion.
	 * @author	Polly Jex (pollyjex@emc2zen.com.ar)
	 */
	public class AbstractIterator extends CoreObject implements IIterator {
		
		/**
		 * Registro.
		 */
		private var _record:Array;
		
		/**
		 * Puntero de lectura.
		 */
		private var _index:int;
		
		/**
		 * Constructor.
		 * @param	value	Coleccion a iterar.
		 * @return	AbstractIterator
		 */
		public function AbstractIterator( value:Array ) {
			
			this._record = value;
			this.reset();
			
		}
		
		/**
		 * Verifica si existe un elemento posterior al actual en la coleccion.
		 * @return	Boolean
		 */
		public function hasNext():Boolean {
			
			return new Boolean( this.index() < ( this.size() - 1 ) );
			
		}
		
		/**
		 * Retorna la posicion actual del puntero.
		 * @return	int
		 */
		public function index():int {
			
			return this._index;
			
		}
		
		/**
		 * Retorna el valor del siguiente elemento de la coleccion.
		 * @return	*
		 */
		public function next():* {
			
			return this.ResolveNext();
			
		}
		
		/**
		 * Retorna el valor del siguiente elemento en la coleccion, pero no altera el puntero de lectura.
		 * @return	*
		 */
		public function peek():* {
			
			return this.ResolvePeek();
			
		}
		
		/**
		 * Remueve y retorna el valor del ultimo elemento retornado de la coleccion.
		 * @return	*
		 */
		public function remove():* {
			
			return this._record.splice( this._index --, 1 );
			
		}
		
		/**
		 * Reseta el puntero de lectura.
		 * @return	void
		 */
		public function reset():void {
			
			this._index = -1;
			
		}
		
		/**
		 * Mueve el puntero de lectura al valor especifico y retorna su valor.
		 * @param	value	Posicion del puntero.
		 * @return	*
		 */
		public function seek( value:int ):* {
			
			this.ResolveSeek( value );
			return this.value();
			
		}
		
		/**
		 * Retorna el numero de datos en la coleccion.
		 * @return	uint
		 */
		public function size():uint {
			
			return this._record.length;
			
		}	
		
		/**
		 * Retorna una cadena con la representacion de la instancia de la clase.
		 * @return	String
		 */
		public override function toString():String {
			
			return new String( "[class " + ClassUtil.shortName( this ) + " size=\"" + this.size() + "\" index=\"" + this.index() + "\"]" );
			
		}
		
		/**
		 * Retorna el valor asociado a la posicion actual del puntero.
		 * @return	*
		 */
		public function value():* {
			
			return this._record[ this.index() ];
			
		}
		
		/**
		 * Resuelve el valor del siguiente elemento de la coleccion.
		 * @return	*
		 */
		protected function ResolveNext():* {
			
			return this._record[ ++ this._index ];
			
		}
		
		/**
		 * Resuelve el valor del siguiente elemento en la coleccion, pero no altera el puntero de lectura.
		 * @return	*
		 */
		protected function ResolvePeek():* {
			
			return this._record[ this.index() + 1 ];
			
		}		
		
		/**
		 * Resuelve el movimiento del puntero de lectura al valor especifico.
		 * @param	value	Posicion del puntero.	
		 * @return	void
		 */
		protected function ResolveSeek( value:int ):void {
			
			this._index = NumberUtil.toLimit( value - 1, -1, this.size() - 1 );
			
		}
		
	}
	
}
